<?xml version="1.0"?>
<!DOCTYPE xsl:stylesheet [
  <!ENTITY rdf "http://www.w3.org/1999/02/22-rdf-syntax-ns#">
  <!ENTITY rdfs "http://www.w3.org/2000/01/rdf-schema#">
  <!ENTITY dc "http://purl.org/dc/elements/1.1/">
  <!ENTITY dcterms "http://purl.org/dc/terms/">
  <!ENTITY foaf "http://xmlns.com/foaf/0.1/">
  <!ENTITY a "http://xmlns.heustics.org/schema/myelin/0.1/">
  <!ENTITY m "http://xmlns.heustics.org/ontology/myelin/0.1/">
]>
<xsl:stylesheet version="1.0"
  xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
  xmlns:f="http://xmlns.4suite.org/ext"
  xmlns:l="http://xmlns.heustics.org/schema/i18n/0.1/"
  xmlns:a="http://xmlns.heustics.org/schema/myelin/0.1/"
  xmlns:o="http://xmlns.heustics.org/schema/myelin/0.1/DomainProfile#new" 
  extension-element-prefixes="f"
  exclude-result-prefixes="xsl a o l"
>   
  <xsl:variable name="o:l10n" select="a:get-l10n('&a;DomainProfile#new')"/>
  <xsl:variable name="o:titles-l10n" select="$o:l10n/l:context[@name='titles']/*"/>
  <xsl:variable name="o:params-l10n" select="$o:l10n/l:context[@name='params']/*"/>
  <xsl:variable name="o:hints-l10n" select="$o:l10n/l:context[@name='param-hints']/*"/>
    
  <xsl:template match="a:title[@key='new-domain-profile']" mode="a:head-content">
    <xsl:call-template name="a:head-title">
      <xsl:with-param name="title" select="$o:titles-l10n[1]"/>
    </xsl:call-template>
  </xsl:template>
  
  <xsl:template match="a:title[@key='new-domain-profile']" mode="a:page-header">
    <xsl:call-template name="a:page-header-title">
      <xsl:with-param name="title" select="$o:titles-l10n[1]"/>
    </xsl:call-template>
  </xsl:template>       
  
  <xsl:template match="a:form[@mode='unavailable']">
    <xsl:param name="status" select="@status"/>
    <xsl:param name="reason" select="$o:l10n/l:context[@name='unavailable-status']/*[@key=$status]"/>
    <xsl:choose>
      <xsl:when test="$reason">
        <xsl:call-template name="a:unavailable-form">
          <xsl:with-param name="status" select="$status"/>
          <xsl:with-param name="reason" select="$reason"/>
          <xsl:with-param name="title" select="$o:l10n/l:context[@name='unavailable-status-titles']/*[@key=$status]"/>
        </xsl:call-template>  
      </xsl:when>
      <xsl:otherwise>
        <f:apply-imports>
          <xsl:with-param name="status" select="$status"/>
        </f:apply-imports>
      </xsl:otherwise>
    </xsl:choose>
    
  </xsl:template>  
  
  <xsl:template match="a:nick" mode="a:input-show" name="o:nick-input"> 
    <xsl:param name="params-l10n" select="$o:params-l10n"/>
    <xsl:param name="hints-l10n" select="$o:hints-l10n"/>
    <div class='input text nick'>  
      <div class="key">        
        <div>
          <xsl:call-template name="a:label-class-attr"/>
          <span><xsl:value-of select="$params-l10n[@key='nick']"/></span>
          <div class='spacer'/>
        </div>   
        <div class='hint'>
          <xsl:call-template name="a:input-hints">
            <xsl:with-param name="hint" select="$hints-l10n[@key='nick']"/>
          </xsl:call-template>
          <div class='spacer'/>
        </div>     
        <div class="spacer"/>
      </div>      
      <div class='value'>    
        <div class="entry">
          <xsl:variable name="opts" select="a:options"/>
          <xsl:choose>
            <xsl:when test="$opts/@count &gt; 1">
              <select name="nick">          
                <xsl:apply-templates select="$opts/*" mode="a:single-select-label">
                  <xsl:with-param name="selected" select="@selected"/>     
                </xsl:apply-templates>
              </select>    
            </xsl:when>
            <xsl:when test="$opts/@count = 1 and $opts/*[1]/@label">
              <span>
                <xsl:value-of select="$opts/*[1]/@label"/>
              </span>
            </xsl:when>
            <xsl:otherwise>
              <span>
                <xsl:value-of select="$o:l10n/l:gentext[@key='none-available']"/>
              </span>
            </xsl:otherwise>
          </xsl:choose>
          <div class="spacer"/>
        </div>
        <div class="spacer"/>
      </div>
      <div class='spacer'/>
    </div>
  </xsl:template>
  
</xsl:stylesheet>

